```
> restart; with(linalg):
```

Warning, the protected names norm and trace have been redefined and unprotected

In class we (are going to) analyze an finite wing with a chord that varies from 6' at the root to 2 ft at the tip. The span is 40 ft, the angle of attack is fixed at 6 degrees at all locations as is alpha lzero at -2.5 degrees. Below is a maple notebook to calculate the 2 term series with theta1 = pi/2 and theta2 = pi/4. We will work it by hand as well as review this notebook in class but I figured those that wanted to could get started on the HW early rather than leaving it for next week when you should be getting ready for the exam. You will be able to use Maple on the exam including a sample notebook with all of these commands included (more on this later).
First define all of your constant and spanwise varying dimensions and calculate things like S, AR,

```
> b:= 40;  c_of_y:=evalf(6+4*y/(b/2));
  alpha_of_y:=evalf((6)/180*Pi);
  alpha_lzero_of_y:=evalf((-2.5)/180*Pi);S:=2
  *int(c_of_y,y=-b/2..0);AR:=b^2/S;
```

$$b := 40$$

$$c\_of\_y := 6. + 0.2000000000\, y$$

$$alpha\_of\_y := 0.1047197551$$

$$alpha\_lzero\_of\_y := -0.04363323131$$

$$S := 160.$$

$$AR := 10.00000000$$

Convert the y (spanwise) variation to theta variation by substituting the map

```
> c:=subs(y=-b/2*cos(theta),c_of_y);
  alpha:=subs(y=-b/2*cos(theta),alpha_of_y);
  alpha_lzero:=subs(y=-b/2*cos(theta),alpha_l
  zero_of_y); evalf(subs(theta=Pi/4,
  c));evalf(cos(Pi/4));
```

$$c := 6. - 4.000000000\cos(\theta)$$

$$\alpha := 0.1047197551$$

$$alpha\_lzero := -0.04363323131$$

$$3.171572876$$

$$0.7071067810$$

Decide how many terms to use and create your series expansion of the various alpha terms (I call this the theta_eq since it is valid for all theta).

```
> nterms:=2;
```

$$nterms := 2$$

```
> theta_eq:=2*b/(Pi*c)*sum('A[2*n-1]*sin((2*n
  -1)*theta)','n'=1..nterms)+alpha_lzero+1/si
  n(theta)*sum('(2*n-1)*A[2*n-1]*sin((2*n-1)*
  theta)','n'=1..nterms)=alpha;
```

$$theta\_eq := 80\,\frac{A_1\sin(\theta) + A_3\sin(3\,\theta)}{\pi\,(6. - 4.000000000\cos(\theta))} - 0.04363323131 + \frac{A_1\sin(\theta) + 3\,A_3\sin(3\,\theta)}{\sin(\theta)} = 0.1047197551$$

Since we took 2 terms we need to evaluate the equation at two locations to get 2 equations.

```
> eq1:=evalf(subs(theta=Pi/2,theta_eq));
  eq2:=evalf(subs(theta=Pi/4,theta_eq));
```

$$eq1 := 5.244131814\, A_1 - 7.244131814\, A_3 - 0.04363323131 = 0.1047197551$$

$$eq2 := 6.677412130\, A_1 + 8.677412130\, A_3 - 0.04363323131 = 0.1047197551$$

solve the equations and then calculate the required aerodynamic quantities

```
> soln:=solve({eq1,eq2},{A[1],A[3]});
  assign(soln); A[1],A[3];
```

$$soln := \{A_1 = 0.02516052748, A_3 = -0.002264986920\}$$

$$0.02516052748, -0.002264986920$$

```
> CL2:=evalf(Pi*AR*A[1]);
```

$$CL2 := 0.7904412830$$

```
> delta2:=sum('(2*n-1)*(A[2*n-1]/A[1])^2','n'
  =2..2); effic2:=evalf(1/(1+delta2));
```

$$\delta 2 := 0.02431157908$$

$$effic2 := 0.9762654455$$

```
> CDi2:=evalf(CL2^2/Pi/AR*(1+delta2));
```

$$CDi2 := 0.02037142635$$